<?xml version="1.0" encoding="UTF-8"?>
<xsl:stylesheet version="2.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
	xmlns:xs="http://www.w3.org/2001/XMLSchema"
	xmlns:f="livcos.org/ns/xslt/function"
  exclude-result-prefixes="f xs">

  <xsl:import href="../../style/base.xsl" />
  
	<!-- resolve relative ref attributes (false by default) -->
	<xsl:variable name="resolveObjUris" select="false()" />

  <!-- escaped output -->
  
  <xsl:function name="f:escape">
    <xsl:param name="text"/>
    <xsl:value-of
      select="replace(replace(replace($text,'&amp;','&amp;amp;'),'&lt;','&amp;lt;'),'&gt;','&amp;gt;')" />
  </xsl:function>
  
  <xsl:template match="text()" mode="escape root_escape">
    <xsl:value-of select="f:escape(.)" />
  </xsl:template>

  <xsl:template match="@*" mode="escape">
    <xsl:text> </xsl:text>
    <xsl:value-of select="name(.)" />
    <xsl:text>="</xsl:text>
    <xsl:value-of select="f:escape(.)" />
    <xsl:text>"</xsl:text>
  </xsl:template>

  <xsl:template match="*" mode="escape">
    <xsl:param name="root"/>
    <xsl:text><![CDATA[<]]></xsl:text>
    <xsl:value-of select="name(.)" />
    <xsl:apply-templates select="@*" mode="escape"/>
    <xsl:for-each select="namespace::*[local-name(.) != 'xml']">
      <xsl:variable name="prefix" select="local-name(.)"/>
      <xsl:variable name="nsUri" select="."/>
      <xsl:if test="not(../../namespace::*[(local-name(.) = $prefix) and (. = $nsUri)]) or $root">
        <xsl:text> xmlns</xsl:text>
        <xsl:if test="$prefix != ''">:</xsl:if>
        <xsl:value-of select="$prefix" />="<xsl:value-of select="$nsUri" />
        <xsl:text>"</xsl:text>
      </xsl:if>
    </xsl:for-each>
    <xsl:choose>
      <xsl:when test="node()">
        <xsl:text><![CDATA[>]]></xsl:text>
        <xsl:apply-templates select="node()" mode="escape"/>
        <xsl:text><![CDATA[<]]>/</xsl:text>
        <xsl:value-of select="name(.)" />
        <xsl:text><![CDATA[>]]></xsl:text>
      </xsl:when>
      <xsl:otherwise>
        <xsl:text>/<![CDATA[>]]></xsl:text>
      </xsl:otherwise>
    </xsl:choose>
  </xsl:template>

  <xsl:template match="*" mode="root_escape">
    <xsl:apply-templates select="." mode="escape">
      <xsl:with-param name="root" select="1"/>
    </xsl:apply-templates>
  </xsl:template>


  <!-- link URL converters -->

	<xsl:template name="ref2Url">
    <xsl:param name="ref"/>
		<xsl:param name="default_extension"/>
		<xsl:param name="default_view"/>
		<xsl:variable name="_uri" select="f:uri($ref)"/>
  	<xsl:choose>
  		<xsl:when test="starts-with($ref, '#')">
  			<xsl:value-of select="$ref"/>
  		</xsl:when>
  		<xsl:when test="starts-with($ref, 'javascript:')">
  			<xsl:value-of select="$ref"/>
  		</xsl:when>
  		<xsl:when test="not(f:isValidUrl($ref))">
  			<xsl:value-of select="$ref"/>
  		</xsl:when>
  		<xsl:when test="$_uri/scheme">
  			<xsl:value-of select="$ref"/>
  		</xsl:when>
  		<xsl:when test="normalize-space($ref) != ''">
				<xsl:variable name="p">
					<xsl:call-template name="objUri2Url">
						<xsl:with-param name="objUri" select="$_uri/path"/>
						<xsl:with-param name="default_extension" select="$default_extension"/>
					</xsl:call-template>
				</xsl:variable>
				<xsl:choose>
					<xsl:when test="$default_view and ends-with($p, $default_extension)">
						<xsl:value-of select="replace($p, '/data/', concat('/', $default_view, '/'))"/>
					</xsl:when>
					<xsl:otherwise>
						<xsl:value-of select="$p"/>
					</xsl:otherwise>
				</xsl:choose>
				<xsl:if test="$_uri/query">?<xsl:value-of select="$_uri/query"/></xsl:if>
				<xsl:if test="$_uri/fragment">#<xsl:value-of select="$_uri/fragment"/></xsl:if>
			</xsl:when>
      <xsl:otherwise>
        <xsl:text>#</xsl:text>
      </xsl:otherwise>
    </xsl:choose>
  </xsl:template>

  <xsl:template name="objUri2Url">
    <xsl:param name="objUri"/>
    <xsl:param name="default_extension"/>
    <xsl:choose>
      <xsl:when test="starts-with($objUri, '/')">
        <xsl:value-of select="$toRootPath"/>
        <xsl:value-of select="$objUri" />
      </xsl:when>
      <xsl:when test="$resolveObjUris">
        <xsl:value-of select="$toRootPath"/>
        <xsl:value-of select="resolve-uri($objUri, f:object-id(.))" />
      </xsl:when>
      <xsl:otherwise>
        <xsl:value-of select="$objUri" />
      </xsl:otherwise>
    </xsl:choose>
    <xsl:if test="not(matches($objUri,'\.[^/]+$')) and not(ends-with($objUri, '/'))">
      <xsl:value-of select="$default_extension"/>
    </xsl:if>
  </xsl:template>

</xsl:stylesheet>
